
News  

US Training Days 
Our US training days will take place on the 13th and 14th November at the Doubletree   
Hotel, Washington DC, prior to the Society for Neuroscience annual meeting. The training 
days are suitable for both existing and prospective users of Spike2 and Signal and 
include introductions to the software along with sessions on script writing, stimulus 
generation and on-line control. Our popular workshop thread, where you can discuss 
your questions and requirements with a CED engineer or programmer in a small group, 
will also be available. 

If you or any of your colleagues are interested in attending, please follow the links to view 
the training day program and complete the registration form. 

Meetings and events 
3rd International Conference on TMS and Direct Current Stimulation 
University of Gottingen 
Gottingen, Germany  
October 1st – 4th 2008 

Biomedical Engineering Society 
Renaissance Grand Hotel 
St. Louis, MO, USA 
October 2nd – 4th 2008 

6th Chinese National Conference on Cardiovascular and Kidney Physiology 
ChengDu Grand Hotel 
ChengDu City, Sichuan Province, China 
October 10th – 13th 2008 

Beijing Joint Conference of Physiological Sciences 
Beijing Xijiao Hotel 
Beijing, Haidian District, China 
October 19th – 22nd 2008 

Society for Neuroscience 
Washington Convention Center 
Washington, DC, USA 
November 15th – 19th 2008 

Latest versions of Spike2 and Signal 
Updates for Spike2 and Signal are available from the CED downloads page, or by 
clicking on the links in the table below. Demonstration versions of the latest software are 
also available.   

 Spike2 downloads Signal downloads 
Spike2 version 6.09 Signal version 4.02 

Spike2 version 5.20 Signal version 3.10 

Spike2 demo Signal demo 

Q. Is there any way to classify spike collisions that occur in WaveMark data? 

A. There is a collision analysis mode available from the Analyse menu in the Edit 
WaveMark dialog that can be used to search for pairs of templates whose sum is 
most similar to the spike collision. These collisions can then be classified 
accordingly. 

 
Collision analysis mode dialog 

The main oscilloscope type display shows the current spike, which is overlaid 
with the best match combination of templates. You can select which templates to 
include in the matching procedure by clicking the  (include) or  (exclude) 
button in the template display area and assign the best match template codes to 
the current spike using the Copy codes button in the dialog. 

If the WaveMark channel being analysed is a memory channel and the current 
spike can be usefully separated, then the option to ‘split spike’ will also be 
available. 

Further details and information regarding the matching algorithm can be found in 
the Spike2 on-line help. 

   

 

 

Q. Is it possible to enter your own text for the display of a vertical cursor in Signal? 

A. The displayed cursor label can be changed by right-clicking on the cursor and 
following the Cursor option in the context menu. This leads to a list of options for 
the selected cursor which includes Set label... 

 
Cursor label dialog 

 Options for the display includes User defined where you can enter a combination 
of text and selected values such as position, cursor number and channel value. 
The settings entered above are used for the display of the cursor on the MEP 
data below. 

CED
You can also manually assign codes to spikes by double clicking the code value in the template window. This can be used, for example, to set two templates to the same code

CED
You can also set active modes and copy cursor positions and values by right-clicking on a vertical cursor



 
MEP data with user defined cursor label 

Scripts: Spike2  
Q. I am running a force feedback experiment and would like to display the current 

waveform value in a separate window, perhaps as a histogram type bar. Is there 
a script available for this? 

A. The attached script, BarDisplay.s2s, is an example of displaying an on-line 
channel value in a separate window for use in a force feedback experiment, for 
example. The script contains an idle routine that updates the bar display in 
response to incoming waveform data in channel 1. 

Scripts: Signal  
Q. Is there a batch export script available that will export all data files in a directory 

to text files? 

A. The attached script, BatchExport.sgs, allows the user to select a directory of 
files for export to text. This script could be modified to export data to other 
supported data types such as metafile or MATLAB data. 

Did you know…? 
The Power1401 mkII can be set to use ± 5 or ±10 Volt range using the Try1401 
application software. 

Recent questions   
Q. I am interested in R-R interval analysis for heart rate variability studies. Using the 

examples from previous newsletters (enews #28 and #29) I have managed to 
generate a data channel measuring the R-R interval values and plot them as 
RealMark data. I would like to perform power spectrum analysis on this channel 
but this option doesn’t seem to be available. 

A. The Power spectrum analysis is only valid for a waveform or RealWave (32-bit 
waveform) channel, but you can convert a RealMark channel to a RealWave 
channel using a Virtual channel in Spike2. When you create a new Virtual 
Channel you can enter an expression that creates a waveform channel using the 
RealMark data values.  

opening the embedded 
scripts in this newsletter 
please let us know. 

Back to top 
Right-click the script icon 
and save to disk. 

If you have any problems 
opening the embedded 
scripts in this newsletter 
please let us know. 

Back to top

Back to top


'$BarDisplay|Generates a single bin result view and displalys the current level on a waveform channel in this as a bar.



'The script is a 'work in progress' and is offered without guarantees.

'You must test it to see whether it is suitable for your application.



'CED 29/08/08

var ch% := 1;                        'Waveform channel to monitor



var data%;						    'Handle of new data file (Spike2 sets it to 0)

var sTime;							'The last time we looked at the idle routine

var showLevel%;						'Result view to show level on Channel 1

var level;								'Current level of channel 1	

 

var cScale,cOffset;                 'Scale and offset of source waveform channel



HideAll();                       'Hide all unwanted bars

ToolbarVisible(1);			    'Make toolbar visible always

New%();									'Set up new sampling window

DoToolbar();							'Do the toolbar

RestoreAll();                    'Restore all bars

Halt;



func New%()							'New sampling window

View(LogHandle());					'Make log view the current view

EditSelectAll();						'Select all text in log view

EditClear();							'Delete it

Window(0,80,100,100);				'Display it at the bottom of the screen

WindowVisible(1);					'Make it visbible



if data%>0 then						'If there is already a data view open...

  View(data%);							'...close it

  FileClose();                   'No point setting data% to 0...

endif;



data%:=FileNew(0,1);					'Open a new data file for sampling

if data%<0 then Message("Unable to open new data file");Halt() endif;

DrawMode(-1,2);						'Set event draw mode to lines

Window(0,0,80,100);					'Make data window in top bit of screen

XRange(0,10);

cScale := ChanScale(ch%);

cOffset := ChanOffset(ch%);





'--------------Setup result view--------------

showLevel% := SetResult(1,1,0,"Level","");          'Create a new result view for the bar display

XRange(0,1);

YRange(1,((-5 * cScale)+ cOffset),(5*cScale)+cOffset);      'Set Y range and channel title to match source waveform channel

ChanTitle$(1,View(data%).ChanTitle$(ch%));          'Copy title from waveform channel

Window(80,0,100,100);

FrontView(showLevel%);

'---------------------------------------------



FrontView(data%);						'Bring the data view to the front

ToolbarEnable(3,0);					'Disable "Sample stop" button

ToolbarEnable(2,1);					'Disable "Sample stop" button

ToolbarText("Press SAMPLE START to commence sampling");

return 1;

end;



proc DoToolbar()

ToolbarSet(0,"",Idle%);				'Call Idle%() whenever there is free time

ToolbarSet(1,"&Quit",Quit%);		'Set up toolbar buttons

ToolbarSet(2,"&Sample start", Start%);

ToolbarSet(3,"&Sample stop", Stop%);

ToolbarSet(4,"&New file", New%);

ToolbarEnable(3,0);					'Disable "Sample stop" button

Toolbar("Press SAMPLE START to commence sampling", 1023);'Wait here until quit

end;



func Quit%()							'If "Quit" is pressed

SampleStop();							'Stop sampling

return 0;								'leave toolbar

end;



func Start%()						'If "Start" is pressed

SampleStart();						'Start sampling

ToolbarEnable(4,0);					'Disable "New file" button

ToolbarEnable(3,1);					'Enable "Sample stop" button

ToolbarEnable(2,0);					'Disable "Sample start" button

ToolbarEnable(1,0);					'Disable "Quit" button

ToolbarText("Press SAMPLE STOP to stop sampling");

return 1;								'Stay with toolbar

end;



func Stop%()							'If "Stop" is pressed

SampleStop();						'Stop sampling

if (data%>0) and                 'if no data in file, stop will close it

   (ViewKind(data%) <> 0) then data% := 0 endif;

ToolbarEnable(4,1);					'Enable "New file" button

ToolbarEnable(3,0);					'Disable "Sample stop" button

ToolbarEnable(1,1);					'Enable "Quit" button

ToolbarText("Press FILE NEW to capture more data");

return 1;								'Stay in toolbar

end;



func Idle%()							'Idle routine runs when system has time

var eTime;

if ViewKind(data%)=0 then

   View(data%);

   eTime:=Maxtime();                'Get the latest time 



	if eTime > sTime then

 	  level := ChanValue(ch%,MaxTime()-BinSize(1));					'Get level on channel 1

		ArrConst(View(showLevel%).[:1],level);                      'Set result view bin to level

		DrawAll();                      'Make sure bar display is updated

        sTime:=eTime;           

	endif;

endif;

return 1;								'Stay in toolbar

end;



'You can find this code in the App() on-line help.

var gFloat%[20];              'global for floating window states

proc HideAll()

var i%;

gFloat%[0] := App(-2);        'number of windows

for i% := 1 to gFloat%[0] do  'hide all windows and save state

   gFloat%[i%] := View(App(i%)).WindowVisible(0);

   next;

end



proc RestoreAll()

var i%;

for i% := 1 to gFloat%[0] do  'restore hidden windows

   View(App(i%)).WindowVisible(gFloat%[i%]);

   next;

end





CED
BarDisplay.s2s


'$BatchExport|Script to export all data files in a directory as text files



'The script is a 'work in progress' and is offered without guarantees.

'You must test it to see whether it is suitable for your application.



'CED 29/08/08



var files$[50];		'Declare array to hold up to 50 files

var chans%[32];     'Declare array to hold up to 32 channels

var numFiles%;			'Number of files found

var c%;				'counters to loop through files 

var success%;			'Test if file opened correctly

var name$;          'File name to save as text

var text%;          'Current text file



FilePathSet("",0,"Select directory of files for export to text");		'Prompts user to select directory

numFiles%:= FileList(files$[],0);		'Fill array with up to 50 files and return number of files found

if numFiles%<1 then							'If no files found...

	Message("No data files in directory!");		'Display message and end script

halt;endif;



for c%:=1 to numFiles% do			'Loop through each file in the directory

	success%:=FileOpen(files$[c%-1],0);		'Check it opened correctly



	if success% > 0 then			'If it opened correctly..

		ChanList(chans%[]);		'Get a list of the channels in the file

        name$:=FileName$(3);

		ExportChanFormat(0,1,1,1);      'Export using default settings for data types

        ExportChanFormat(1,0,0,0);

        ExportChanFormat(4,0,0,0);

        ExportChanFormat(5,0,0,0);

        ExportTextFormat(6,8,0,"\"","\t",1,0);  'Default de-limit for text data

        ExportChanList(chans%[]);

        ExportFrameList(-1);

        ExportTimeRange(0,MaxTime());

        EditCopy(4);                        'Copy data as text

        text%:=FileNew(1,1);            'Create new text file

        EditPaste();

        FileSaveAs(name$+".txt",1);     'Save with same name and to same directory

        FileClose(-1);                      'Close text file

	else;

		Message("Could not open file %s",files$[c%]);	'If file not opened, display error message

	endif;

	

next;





     



CED
BatchExport.sgs

http://www.ced.co.uk/img/cednws28.pdf
http://www.ced.co.uk/img/cednws29.pdf


Follow the >> button to the right of the expression field and select Waveform 
from Channel and then select RealMark data item... You can then edit the 
settings for maximum gap and interpolation method to generate a waveform from 
this. 

 
Settings to create a RealWave channel from RealMark data 

User group      
If you have any comments about the newsletter format and content, or wish to un-
subscribe from the mailing list for this newsletter, please notify sales@ced.co.uk. 

 
All Trademarks are acknowledged to be the Trademarks of the registered holders. 

